SSEC/TP-41237

MODIFICATIONS TO THE FCHART/SLR VERSION 2.0 PROGRAM

J.M. Hill

July 1981

prepared for

U.S. DEPARTMENT OF ENERGY
under contract number DE-AC02-79CS30166

SOUTHERN SOLAR ENERGY CENTER
OPERATED WITH SUPPORT FROM THE U.S. DEPARTMENT OF ENERGY
61 PERIMETER PARK ATLANTA, GEORGIA 30341 404 458-8765

DISCLAIMER

SSEC/TP--41237

DE82 009471

MODIFICATIONS TO THE FCHART/SLR VERSION 2.0 PROGRAM

J.M. Hill

July 1981

DISCLAIMER

ABSTRACT

A number of errors have been detected in the FCHART/SLR computer code as it pertains to the thermal performance of passive solar energy systems. Along with minor coding changes, major revisions in the code have been made to improve the computer models used to predict the effects of overhangs on incident solar radiation and the radiation absorbed in attached sunspaces. Modifications to the code were also made to improve the handling of mullions and to reduce the effort required to describe the placement of overhangs. The theoretical basis of these changes, along with the associated alterations to the code, are given. For the cases examined, the program as modified now agrees to within 15% of published LANL passive system performance correlations. This new code has been designated as Version 2.1 and is presently operational at SSEC.

TABLE OF CONTENTS

LIST OF FIGURES

LIST OF LISTINGS

LIST OF APPENDIXES

MODIFICATIONS TO THE FCHART/SLR VERSION 2.0 PROGRAM

INTRODUCTION

The FCHART/SLR program was developed to predict the performance of both active and passive solar energy systems (reference 1). The program is a modification of the FCHART program developed by the University of Wisconsin, a computerized version of the f-chart active solar system correlations. The FCHART/SLR program basically uses the environmental data and radiation processor existent in the original FCHART program and applies this information along with an additional thermal description of the passive structure to the Los Alamos National Laboratory (LANL) Solar Load Ratio (SLR) performance correlations for passive solar systems. The types of passive systems that can be modeled are direct gain, Trombe wall, water wall, and attached sunspaces. The program can be used exclusive of or in conjunction with the active solar system portion of the program. The Northeast Solar Energy Center (NESEC) in conjunction with Total Environmental Action, Inc. (TEA) developed the FCHART/SLR program and has released a user's manual (reference 2) that gives a good description and discussion of the operation of both the active and passive solar systems and their inherent limitations.

Since the implementation of this program at the Southern Solar Energy Center (SSEC), a number of problems with the passive portion of the program were noticed. This led to a systematic examination and verification of the operation of all passive system performance routines in the FCHART/SLR Version 2.0 program. As a result it was found that, as released to SSEC, the FCHART/SLR code did not accurately predict the performance of passive solar systems for all reasonable input requirements when compared to the performance correlations published by LANL. This is especially true for the most recent addition to the program, the performance of attached sunspaces. A number of code revisions have been made at SSEC to reduce the discrepancy between LANL performance predictions and the program results. For all cases run, the present SSEC version of the FCHART/SLR program agrees with published data to within 15%. This degree of confidence is expected for any passive system design conforming to the assumptions published by LANL and obeying normal passive design practices as outlined in Volume 1 of the Passive Solar Design Handbook (PSDH).

This is a report of the problems found to date in and modifications made to the FCHART/SLR code at SSEC. Since each section deals with essentially a separate subject, the main body of the report may appear rather disjointed. However, the nomenclature has been developed as consistently as possible considering the diversity of the subject matter. It is hoped that the included nomenclature list will be of value in reducing the confusion of addressing a number of rather complex problems. A brief description of the major problems encountered is given below with a more detailed derivation of the changes and necessary code revisions given in the Appendices.

PROGRAM REVISIONS

Adjustments of the computer code arise from two types of problems, coding errors and model errors. Coding errors are mistakes in the implementation of a particular computer model due to either mistakes in translating a physical problem into a computer model by the programmer or mistakes in generating the code once it has been developed by the programmer. Such problems are to be expected for a program of the size and complexity of FCHART/SLR. The best cure for any remaining coding errors is extensive use of the program with a constant feedback to the originator of the code.

Model errors are due to problems in the physical model from which the computer code is generated. This is a more serious problem for several reasons. The first is that the nature of any discrepancy between the results given by the program and the desired results is not easily determined. Upon initial identification of a problem area, there may be no way to discern between a model error and a coding error. A full working knowledge of both the code and the theoretical model is required. The second reason is that, once it has been determined that a model error exists, a new computer model must be developed to describe the desired process. This often requires an adaptation to the code that was not foreseen upon its original implementation. The final problem with model errors is that a ready solution may not exist. Theoretical models developed by LANL (references 3 and 4) upon which the passive features of the FCHART/SLR program are based are rather restrictive in scope. In order to provide a generalized approach to the prediction of passive system performance suitable to a computer code, the program uses its own theoretical models to reproduce results from LANL theoretical models. The difference between these two methods of reproducing the same information is open to question and differing interpretations.

An example of these two types of problems is given in Figure 1. In this figure the solid line is the annual Solar Savings Fraction (SSF) versus the Load Collector Ratio (LCR) as determined for an attached sunspace in Atlanta, Georgia based on the LANL type (a) geometry (shown in Appendix C, Figure C-1) and as discussed in Reference 5. Particular care was taken to maintain the same geometry and sunspace thermal loss as given in the published LANL data, since these directly impact the Load Collector Ratio. The data points shown as triangles in Figure 1 show the results obtained using the FCHART/SLR Version 2.0 program as initially received at SSEC. It is noted that the agreement is not considered adequate in this case.

Based on concern over the lack of agreement (illustrated by the triangle shaped data points), a detailed examination was made of the code. This resulted in identification of several coding errors that resulted from the use of implicit integer variable names as real variables. A correction to these problems gave improved results as indicated by the data points shown as squares in Figure 1. This shows better agreement with the correlation trend (solid line), but it is evident that further improvement is needed.

This led to more extensive examination of the code and the theoretical model used. It was concluded that changes were required to improve the accuracy of the radiation processor used in the model. This was done with the result shown by the circular data points plotted in Figure 1. In this particular case the agreement is now quite good.

A summary of the model changes implemented in SSEC's version of FCHART/SLR is given below. Although most were made to improve the accuracy of the program in predicting published results of the LANL SLR method, some were made for convenience of use. Other modifications to the code were made to reduce the program size and run time, which may not be compatible with other operating systems. These changes have no effect on the program operation and will only be mentioned as they apply to any coding changes made to alter the program results. The changes in the code due to model changes are included in the appendices.

Shading

The original shading algorithm in FCHART/SLR version 2.0 was patterned after that in Version 1.0. The constant shading and monthly shading routines operate as before. Because of coding problems, a misrepresentation of the 1/arctangent as arctangent, the autoshade option did not operate properly. A correction of this problem resulted in the shading routine operating the same as that in Version 1.0.

Further modifications were introduced to facilitate the use of the autoshade option. Originally, the user was required to input the fraction of the glazing area shaded at the winter and summer solstices. Several hand calculations were necessary to determine these values from working drawings of the structure to be modeled. Rather than using these two variables, the user is now required to input the separation ratio, and overhang ratio, as used in the PSDH Vol. 2. This requires only the division of three numbers readily available from the working drawings. These changes are shown in Appendix A.

Perhaps the most significant change in the handling of shading in the FCHART/SLR program has been in the application of the shading factor to the incident radiation. Both Versions 1.0 and 2.0 of the FCHART/SLR program applied a calculated shading fraction to the radiation absorbed by a surface as follows:

$$I_{abs} = f_{shade} \, I_{abs,\,ns}$$

Where:

I_{abs} = absorbed radiation

f_{shade} = average monthly non-shaded fraction

$I_{abs,\,ns}$ = radiation that would be absorbed if no shading was present

The problem with this model is that the monthly shading fraction is determined by calculating a shade line on the transparent glazing based on the average monthly sun position. In this case, the shading fraction is only the fraction of the beam radiation incident on the glazing surface, not the total radiation. The program was modified to include the effects of shading on diffuse and reflected components of radiation as follows:

$$I_{abs} = I_{beam}\, f_{shade}\ \tau\alpha_{beam} + \tau\alpha_{dif}(I_{dif} + I_{ref})$$

Where:

I_{abs} is given above,

I_{beam} = the beam radiation,

f_{shade} is given above,

$\tau\alpha_{beam}$ = the transmittance-absorptance product for the beam radiation,

$\tau\alpha_{dif}$ = the transmittance-absorptance product for the diffuse radiation, and

I_{dif} and I_{ref} = the incident diffuse and reflected insolation respectively.

A check of this modification was made by comparing predicted values of incident radiation from the program for various shading values to those obtained from the PSDH Vol. 2. Agreement between the two methods for those months in which heating loads would normally occur was good over a wide range of climatic conditions. Figure 2 shows typical results for Atlanta, Georgia. Coding changes required to implement these changes are shown in Appendix B.

Attached Sunspaces

One of the improvements of Version 2.0 of FCHART/SLR over Version 1.0 is that Version 2.0 incorporates the latest LANL sunspace model. Unfortunately, the attempt to extrapolate the model to include geometries other than the two published by LANL was not totally successful. After extensive examination of the program and the LANL correlation equations, it was found that a major discrepancy existed between radiation values used by the program and those published by LANL.

The essence of this discrepancy was twofold. The first problem was in the radiation processor used to predict the radiation absorbed in a sunspace with insulated end walls. The program as received from NESEC always considered the radiation striking the end walls as absorbed radiation in the sunspace whether or not the end walls were glazed or insulated. A modification of the code to eliminate this problem made a vast improvement in the ability of the program to reproduce the LANL-published estimate of the sunspace performance for geometry (a) with insulated end walls. However, the other geometry types still produced extremely optimistic performance results. This discrepancy was the result of a model error in the sunspace radiation processor. The problem was essentially the inability of the program to account for radiation losses from the attached sunspace.

Subsequent discussions with LANL indicated that the thermal performance equations for the sunspace geometries were based on the published radiation correlations and were thus an integral part of the model. An examination of the PASOLE program used to develop the radiation correlations published by LANL (reference 5) led to changes in the sunspace radiation processor, resulting in a reduction in the discrepancy between absorbed radiation predicted by FCHART/SLR and those published by LANL as shown in Appendix C. At present, the modified FCHART/SLR program is able to predict published sunspace performance to within 15%, as shown in Figures 1 and 3.

It should be noted that LANL has released performance equations based only on two geometry types. Although the FCHART/SLR program has been developed to accept a much wider range of sunspace designs, any design other than those published by LANL cannot at this point be verified. Until further work is done to generalize the geometric effects on sunspace performance so that more geometries can be tested, the results obtained from the sunspace segment of FCHART/SLR should be treated with caution.

Mullion Effects

Both Versions 1.0 and 2.0 of FCHART/SLR have included a factor for each passive system to take into account the fraction of the gross glazing area covered by mullions. The effect of this factor has been to reduce the amount of absorbed radiation by an equivalent fraction. Hand checks at SSEC of Version 1.0 and 2.0 with small mullion factors (on the order of 5%) with results published in the PSDH Vol. 2 originally indicated this to be an appropriate way to account for mullions. A subsequent run of the Version 2.0 program for an SSEC case study with an unusually high mullion factor (23%) produced a much lower annual SSF than expected (9% compared to an expected 30%). Monthly hand calculations using the above handling of mullions led to the same low annual SSF.

At this point, we questioned the use of a mullion factor to simply reduce the absorbed radiation. A mullion was seen to operate as a non-glazed portion of the south facing wall. In this respect its effect on a passive system was two-fold. First, and most obviously, a mullion reduces the transparent glazing surface. This has partially been accounted for by the reduction of the absorbed radiation as previously discussed. It also affects such glazing-area-dependent variables as the LCR. This was not accounted for by the program and was subsequently changed.

A second effect was that, since the mullions do not transmit radiation to the structure yet still act as a thermal loss, they should not be treated as adiabatic south glazing. In this respect, the mullion thermal losses should go to increase the overall Building Load Coefficient (BLC). The FCHART/SLR program was thus modified to increase the BLC by a factor equal to the mullion area times a mullion U-value. A value of 0.5 Btu/hr - ft^2 - ^oF was chosen as a typical heat loss value for the mullions. This modification was used for direct gain, Trombe wall, and water wall systems only, since any mullion thermal losses in a sunspace are already accounted for in the sunspace LCR_s and do not affect the BLC. Overall losses from the solar system as they affect the non-solar degree days are also assumed to be unaffected by the addition of mullions.

With the addition of these modifications to the FCHART/SLR program, the annual SSF for the case study increased from 9 to 27%, much closer to an expected value of 30%.

Coding Errors

A list of the coding errors found at SSEC in implementing FCHART/SLR on the Cybernet is given in Appendix D.

CONCLUDING REMARKS

One purpose of the FCHART/SLR program is to reduce the work required to estimate the thermal performance of passive solar energy systems using the LANL performance correlations. The modifications to the FCHART/SLR computer code made at SSEC has improved the ability of the program to reproduce published LANL performance predictions. Where possible, the accuracy of these changes has been documented in this report. At this point, test runs of the program indicate that, with the revisions, the program will reproduce the LANL performance estimates, given the same design data, to within 15%. This modified code has been designated Version 2.1 and is presently at use at SSEC.

It is important to note that the program should only be used to facilitate the use of the LANL design procedures. A prediction of passive system performance occurs only when the LANL design method is applied properly through the program to the particular solar system design. This requires a thorough understanding of the assumptions and limitations of the LANL passive solar design methodology to assure proper results from the program.

One of the unfortunate aspects of the flexibility of the FCHART/SLR program is that it provides a simple means of extrapolating the LANL design procedures beyond their established bounds of applicability. Although the result from the program may prove valid in such an extrapolation, there is no a priori guarantee of the results. Any passive system design requiring thermal analysis that does not fit the published LANL model constraints should be examined with more sophisticated thermal network software such as PASOLE, SUNCAT, CALPAS, or DEROB.

The FCHART/SLR version 2.0 program also contains options allowing the modeling of active solar systems and the economic evaluation of both active and passive solar systems. As of yet, no effort at SSEC has been made to verify the operational status or accuracy of these options.



FIGURE 1 LANL SSF sunspace correlation (solid line) for sunspace geometry (a) insulated end walls in Atlanta, Georgia compared to FCHART/SLR program predicted values as received (triangle), with coding errors eliminated (square), and as now operating after modifications to the radiation model (open circles).





FIGURE 2 Comparison of modified and original FCHART/SLR shading models with published LANL correlations. Based on 100 sq. ft. of vertical glazing. All results based on the same base (non-shading) radiation values.



FIGURE 3 Comparison of published LANL attached sunspace correlations (solid lines) with FCHART/SLR predictions (circles) for three geometry types. Runs are for Atlanta, Georgia. Geometry types are shown in Appendix C, Figure C-1.

NOMENCLATURE LIST

abs_j	absorptivity of surface j
A_s	glazing area of sloped surface of LANL sunspace geometry (b)
A_v	glazing area of vertical surface of LANL sunspace geometry (b)
BLC	Building Load Coefficient as defined in Vol. 2 of the Passive Solar Design Handbook
$F_{j \rightarrow k}$	view factor from surface j to surface k
FS	summer shading fraction used by FCHART/SLR version 2.0
f_{shade}	average monthly unshaded fraction of an equitorial-facing glazing surface
FW	winter shading fraction used by FCHART/SLR version 2.0
H	height of a vertical glazing surface
I	instantaneously absorbed solar radiation in a sunspace
$\bar{I}$	absorbed solar radiation in a sunspace over a month
I_{abs}	solar radiation absorbed in a passive solar system with shading present
$I_{abs,\ ns}$	solar radiation absorbed in a passive solar system with no shading present
I_{beam}	beam solar radiation incident as a glazing surface
I_{dif}	diffuse solar radiation incident on a glazing surface
I_{inc_k}	incident solar radiation on transparent surface k
I_k	solar radiation absorbed in a sunspace due to radiation incident on surface k
I_{ref}	reflected solar radiation incident on a glazing surface
I_{tran_k}	solar radiation transmitted through transparent surface k
LAT	Latitude in Degrees
LCR	Load Collector Ratio as defined in Vol. 2 of the Passive Solar Design Handbook
LCR	Load Collector Ratio for a sunspace as defined in reference 5

O	overhang distance for a window overhang
OH	ratio of overhang distance to glazing height for a shaded window
S	vertical separation distance between overhang and top of window for a shaded window
SLR	Solar Load Ratio as defined in Vol. 2 of the Passive Solar Design Handbook
SP	ratio of separation distance to glazing height for a shaded window
SSF	Solar Savings Fraction
XS	shade line position measured from the top of a window of summer solstice
XW	shade line position measured from the top of a window at the winter solstice

Greek symbols

α	surface absorptance
$\tau\alpha$	transmittance absorptance product
$\tau\alpha_{beam}$	transmittance absorptance product for beam radiation
$\tau\alpha_{dif}$	transmittance absorptance product for diffuse (and isotropically reflected) radiation

REFERENCES

1. "F-CHART/SLR: An Interactive Program for Determining the Performance of Active and Passive Solar Buildings"; P. Sullivan and W. Wright, Proceedings of the 1980 Annual Meeting of AS-ISES, p. 875, June 1980, LA-UR-80-1482

2. User's Manual: FCHART/SLR Version 2.0, Northeast Solar Energy Center, 29 Jan 1981

3. Passive Solar Design Handbook, Vol. 1 & 2; J.D. Balcomb, et al.; January 1980; DOE/CS-0127/2

4. "Design and Analysis of Direct Solar Gain Buildings;" W.O. Wray, Los Alamos National Laboratory, Rough Draft

5. "Performance Estimates for Attached-Sunspace Passive Solar Heated Buildings;" R.D. McFarland and R.W. Jones, Proceedings of the 1980 Annual Meeting of AS-ISES, June 1980, LA-UR-80-1482

6. A Design Procedure for Solar Heating Systems; S.A. Klein, PhD Thesis, University of Wisc., 1976

APPENDIX A

Use of Overhang and Separation Ratios in the Autoshade Algorithm

The original FCHART/SLR program used two parameters in its autoshade model, the fraction of the glazed surface shaded from direct radiation during the winter solstice, FW; and the fraction of the glazed surface shaded from direct radiation during the summer solstice, FS. It was tacitly assumed that the glazing is south facing. It was desired to use the overhang, OH, and separation ratios, SP, as defined in the PSDH Vol. 2 rather than FW and FS. This required the development of a functional relationship between the two sets of variables. Figure A-1 shows a typical shading geometry and the physical meaning of the variables FW, FS, OH, and SP.

During the winter solstice at solar noon, XW, S, and O can be related as follows:

$$\tan(LAT + 23.45^{o}) = O/(XW + S), \qquad \text{(A-1)}$$

where LAT is the latitude of the glazing surface.

Solving for XW in terms of O and S gives:

$$XW = O/\tan(LAT + 23.45^{o}) - S. \qquad \text{(A-2)}$$

Now, for FW = XW/H, then FW is related to SP and OH by:

$$FW = OH/\tan(LAT + 23.45) - SP. \qquad \text{(A-3)}$$

A similar derivation can be applied to summer shading to yield FS in terms of OH and SP as:

$$FS = OH/\tan(LAT - 23.45) - SP. \qquad \text{(A-4)}$$

The changes incorporated into the computer coding and data file due to the alteration of these variables are shown in Table A-1.

It should be noted that this shading model (and the original auto-shade model) only applies to zero-azimuth-south-facing glazing. For small changes in azimuth angle (plus or minus 15 degrees), the effects of a non-zero azimuth angle are small, and the model provides acceptable results. For large non-south swings, hand calculations should be made of the monthly shading fraction (weighted by solar flux) and the monthly option used.



Figure A-1 Shading Geometry

LISTING A-1: Changes in FCHART/SLR coding (subroutine shade) and data file due to implementation of shading and overhang ratios.

Original coding and typical data file listing.

```
      SHAD1=PLIST(ISYS,11)
      SHAD2=PLIST(ISYS,12)
   19 SHADNG=SHAD1+(SHAD2-SHAD1)*FALTV
      GOTO 40
```

```
P1 11 JANUARY SHADING/YEAR-ROUND SHADING/AUTO 1....              0.00
P1 12 FEBRUARY SHADING/AUTO 2.......................             0.00
```

Modified coding and data file.

```
C..4-30-81 JMH  SHADE MODIFIED TO ACCEPT OVERHANG AND SEPRATION VALUES
C..AS PER PSHB VOL 2.
C..PLIST(ISYS,11) NOW EQUAL TO OVERHANG RATIO
C..PLIST(ISYS,12) NOW EQUAL TO SEPARATION RATIO
      SHAD1=PLIST(ISYS,11)/TAN((ALAT+23.45)*D2R)-PLIST(ISYS,12)
      SHAD2=PLIST(ISYS,11)/TAN((ALAT-23.45)*D2R)-PLIST(ISYS,12)
   19 SHADNG=SHAD1+(SHAD2-SHAD1)*FALTV
      GOTO 40
```

```
P1 11 JANUARY SHADING/YEAR-ROUND SHADING/OVERHANG..              0.00
P1 12 FEBRUARY SHADING/SEPARATION..................              0.00
```

Note: Changes to the data file must be made for all systems in both the parameter description field and default input values.

APPENDIX B

Application of Shading Value to Absorbed Radiation

An explanation of the modification of the absorbed radiation due to shading has been given in the main body of the report. Implementation of this modification effects two subprograms in FCHART/SLR, subroutine PCALSO and subfunction RBAR. A common block "TB" was set up to pass the necessary variables out of RBAR into PCALSO. The addition of a separate common block was used out of convenience and may be changed to blank common if required. The following Listing B-1 shows the coding modifications to PCALSO. Listing B-2 gives the modifications to RBAR.

LISTING B-1 Modifications to subroutine PCALSO

Original coding

```
      NG=2
      ABSORF=FLIST(ISYS,5)
      Z=AREA*RBAR(MONTH,XLATTI,TILT,AZIMUT,H(MONTH), RHO(MONTH),NG,AIM,T
     1ARAT,TABAR,1)
      IF (ITYPE .NE. 4) GOTO 10
C  HERE TO COMPUTE Z FOR SUNSPACES. WE DO IT BY RUNNING RBAR THREE
C  TIMES: ONCE FOR EACH SURFACE (TILTED, VERTICAL AND ENDWALLS). SINCE
C  THERE ARE 2 ENDWALLS, WE COUNT THEM TWICE. THE Z FROM ABOVE IS USEFUL
C  HAPPENS TO BE THAT OF THE TILTED PORTION.
      TILT=90.0
      AREA=FLIST(ISYS,24)
      Z1=AREA*RBAR(MONTH,XLATTI,TILT,AZIMUT,H(MONTH), RHO(MONTH),NG,AIM,
     1TARAT,TABAR,1)
      TILT=90.0
      AREA=2.0*FLIST(ISYS,25)
      AZIMUT=AZIMUT+90.0
      Z2=AREA*RBAR(MONTH,XLATTI,TILT,AZIMUT,H(MONTH), RHO(MONTH),NG,AIM,
     1TARAT,TABAR,1)
      Z=Z+Z1+Z2
C  COMPUTE THE SHADING VALUE FOR THE SYSTEM. AS A SIDE EFFECT,
C  COMPUTE THE DECLINATION OF THE LOCATION FOR REFLECTANCE
C  CALCULATIONS LATER.
   10 CALL SHADE(ISYS,MONTH,XLATTI,DECL,SHADNG)
C  OPTIONALLY, USE A REFLECTANCE OTHER THAN 1.0 IF A REFLECTOR
C  EXISTS
      IF (IREF .EQ. 2) GOTO 100
      DIF=ALAT-DECL
      REFL=1.0083-.001787*DIF+ .001916*DIF**2-4.031E-5*DIF**3+2.446E-7*D
     1IF**4
      IF (REFL .LT. 1.0) REFL=1.0
  100 SOLARI=Z*H(MONTH)*SHADNG*DAYS(MONTH)*SCONV*(1.0-FLIST(ISYS,4))
      SOLART=SOLARI*TABAR*REFL*ABSORF
      RETURN
      END
```

LISTING B-1 (continued)

Modified coding

```
      COMMON /TB/TABEAM,TADIF,TANORM,DIRINC,DIFINC


      NG=2
      ABSORP=PLIST(ISYS,5)
      CALL SHADE(ISYS,MONTH,XLATTI,DECL,SHADNG)
      R=RBAR(MONTH,XLATTI,TILT,AZIMUT,H(MONTH),RHO(MONTH),NG,AIM,
     1 TARAT,TABAR,1)
      Z=AREA*(DIRINC*SHADNG + DIFINC)
      ZT=AREA*VFS(ISYS)*(DIRINC*SHADNG*TABEAM + DIFINC*TADIF)*TANORM
      IF (ITYPE .NE. 4) GOTO 10
      TILT=90.0
      AREA=PLIST(ISYS,24)
      Z1=AREA*RBAR(MONTH,XLATTI,TILT,AZIMUT,H(MONTH), RHO(MONTH),NG,AIM,
     1TARAT,TABAR1,1)*VFV(ISYS)
C  FOR NON-GLAZED END WALLS, DO NOT ADD TRANSMITTED OR INCIDENT RADIATION
C  ADDED 5-6-81 JMH
      Z2 = 0.0
      TABAR2=0.0
      IF(IFIX(PLIST(ISYS,28)+.5) .NE. 1) GO TO 20
      TILT=90.0
      AREA=2.0*PLIST(ISYS,25)
      AZIMUT=AZIMUT+90.0
      Z2=AREA*RBAR(MONTH,XLATTI,TILT,AZIMUT,H(MONTH), RHO(MONTH),NG,AIM,
     1TARAT,TABAR2,1)
   20 Z=Z+Z1
      ZST=Z1*TABAR1
C  COMPUTE THE SHADING VALUE FOR THE SYSTEM. AS A SIDE EFFECT,
C  COMPUTE THE DECLINATION OF THE LOCATION FOR REFLECTANCE
C  CALCULATIONS LATER.
   10 CONTINUE
C  OPTIONALLY, USE A REFLECTOR FACTOR OTHER THAN 1.0 IF A REFLECTOR
C  EXISTS
      IF (IREF .EQ. 2) GOTO 100
      DIF=ALAT-DECL
      REFL=1.0083-.001787*DIF+ .001916*DIF**2-4.031E-5*DIF**3+2.446E-7*D
     1IF**4
      IF (REFL .LT. 1.0) REFL=1.0
  100 CONST=H(MONTH)*DAYS(MONTH)*SCONV*(1.-PLIST(ISYS,4))
      SOLARI=Z*CONST
      SOLART=ZT*REFL*CONST*ABSORP
C  FOR A GREENHOUSE, ADD THE RADIATION THROUGH END WALLS
      IF(ITYPE.EQ.4)SOLART=SOLART+ZST*CONST*ABSORP
      RETURN
      END
```

Note: Some modifications included are described in Appendix C.

LISTING B-2 Modifications to function RBAR

Original coding

```
C...THE FUNCTION ACOS IS THE INVERSE COSINE
      THETA=ACOS(COSTH)*180./PI
      IF(IFIX(XIAM+.00001).NE.0) GO TO 50
C..THETA IS THE AVERAGE INCIDENCE ANGLE FOR BEAM RADIATION FOR THE
C..MONTH. IT IS ASSUMED TO EQUAL THE INSTANTANEOUS INCIDENCE ANGLE
C..FOR BEAM RADIATION OUTSIDE THE ATMOSPHERE AT 2.5 HR FROM SOLAR
C..NOON ON THE REPRESENTATIVE DAY OF THE MONTH.
      TANORM=TAUALF(NG,0.0,XKL)
      TADIF=TAUALF(NG,60.,XKL)
      TABEAM=TAUALF(NG,THETA,XKL)
      TABAR=TABEAM*(1.-FKT)*RD/RBAR+FKT*(1.+COSS)/2.*TADIF/RBAR  +RHO*(1
     1-COSS)/2.*TADIF/RBAR
      TARAT=TABAR/TANORM
      RETURN
   50 TARB=1.-XIAM*(1./COS(THETA*RDCONV)-1.)
      TARD=1.0-XIAM*(1.0/COS(60.0*RDCONV)-1.0)
      TARAT=(1.0-FKT)*RD/RBAR*TARB+FKT*(1.+COSS)/2.0*TARD/RBAR+RHO*(1.-C
     1OSS)/2.0*TARD/RBAR
      RETURN
      END
```

Modified Coding

```
      COMMON /TB/TABEAM,TADIF,TANORM,DIRINC,DIFINC

C...THE FUNCTION ACOS IS THE INVERSE COSINE
      THETA=ACOS(COSTH)*180./PI
      TANORM=TAUALF(NG,0.0,XKL)
      IF(IFIX(XIAM+.00001).NE.0) GO TO 50
C..THETA IS THE AVERAGE INCIDENCE ANGLE FOR BEAM RADIATION FOR THE
C..MONTH. IT IS ASSUMED TO EQUAL THE INSTANTANEOUS INCIDENCE ANGLE
C..FOR BEAM RADIATION OUTSIDE THE ATMOSPHERE AT 2.5 HR FROM SOLAR
C..NOON ON THE REPRESENTATIVE DAY OF THE MONTH.
      TADIF=TAUALF(NG,60.,XKL)/TANORM
      TABEAM=TAUALF(NG,THETA,XKL)/TANORM
      GO TO 60
   50 TABEAM=1.-XIAM*(1./COS(THETA*RDCONV)-1.)
      TADIF=1.0-XIAM*(1.0/COS(60.0*RDCONV)-1.0)
   60 DIRINC = RD*(1.-FKT)
      DIFINC = FKT*(1.+COSS)/2.+RHO*(1.-COSS)/2.
      TARAT = (TABEAM*DIRINC + TADIF*DIFINC)/RBAR
      TABAR = TARAT*TANORM
      RETURN
      END
```

APPENDIX C

FCHART/SLR Sunspace Radiation Processor

A number of FCHART/SLR runs were made and compared to the radiation correlations released by LANL for the two published geometries shown in Figure C-1. Figures C-2 and C-3 show the differences between the two radiation models for geometry types (a) and (b) both with insulated and glazed ends. Although these are only for Atlanta, Georgia, the results are typical of the differences between the two models for test runs of 15 southern locations encompassing a wide range of climatic conditions. As can be seen, significant differences in the two models can result.

Since the PASOLE program was used to develop the sunspace radiation correlations, an examination was made of the program coding to determine the method by which radiation is absorbed by a sunspace surface. The PASOLE program treats the amount of radiation absorbed by surface "j" due to solar radiation incident on transparent surface "k" as:

$$I_{abs_{k \to j}} = I_{tran_k} F_{k \to j} \, abs_j \qquad \text{(C-1)}$$

where

$I_{abs_{k \to j}}$ is the radiation absorbed by surface j due to solar radiation incident on transparent surface k,

I_{tran_k} is the solar radiation transmitted through transparent surface k,

$F_{k \to j}$ is the view factor from surface k to surface j, and

abs_j is the absorptivity of surface j.

The radiation correlations given in the released technical reports by LANL on attached sunspace use the above equation implicitly in their radiation model, i.e., the radiation correlations were developed from data obtained from PASOLE computer runs. Neglecting internal reflections, the instantaneously absorbed radiation in the sunspace is given by:

$$I = \sum_k \sum_j I_{abs_{k \to j}} \qquad \text{(C-2)}$$

Now, if the absorptivity of each surface is assumed to be approximately a constant, then:

$$\sum_j I_{abs_{k \to j}} = I_{tran_k} \, \alpha \sum_j F_{k \to j} \qquad \text{(C-3)}$$

where: α is the constant absorptivity of each surface and the other terms are as defined above.

Since, for an enclosure with planar surfaces, the sum of all view factors from one surface to all others is equal to one, Equation (3) is equivalent to:

$$I_k = \tau\alpha I_{inc_k} \qquad \text{(C-4)}$$

where: I_k is the radiation absorbed by the sunspace due to incident solar radiation on transparent surface k,

$\tau\alpha$ is an effective transmittance absorptance value of the glazing and sunspace walls, and

I_{inc_k} is the incident solar radiation on transparent surface k.

This gives:

$$I = \sum_k I_k \qquad \text{(C-5)}$$

Time averages of Equations (C-4) and (C-5) over a month are used by the FCHART/SLR program to determine the absorbed radiation in an attached sunspace.

The major problem with using Equations (C-4) and (C-5) instead of Equations (C-1) and (C-2) to find the absorbed radiation is that the absorptances of all j surfaces are not equal in the PASOLE program. Transparent surfaces are given a value of zero while opaque surfaces can be assumed to have some average value, α . Thus it can be seen that a number of terms in Equation (C-2) will be zero (due to a zero value of abs_j in Equation (C-1)) if an attached sunspace has more than one transparent surface. Using Equations (C-4) and (C-5) in this case would overpredict the absorbed radiation.

The attached graphs of comparisons between the LANL radiation correlation based on PASOLE runs and absorbed radiation values obtained from FCHART/SLR runs indicates that an overprediction indeed is the case. In general, the greater the number of glazed surfaces, the greater the disparity between the two radiation models. The only geometry for which good agreement exists between the LANL radiation correlation and the FCHART/SLR program is geometry (a) with insulated end walls. Since this particular geometry has only one glazed surface, Equations (C-1) and (C-2) and Equations (C-4) and (C-5) are equivalent, and, indeed, the two methods produce essentially the same results. Thus, for any attached sunspace geometry containing only a single glazed surface, one can expect reasonable results from the FCHART/SLR program. For those with multiple glazed surfaces the absorbed radiation will be overpredicted by the program yielding an excessively high value of the SSF. The above equations and their differences also give a method of resolving this problem.

It should first be noted that, according to LANL radiation correlations (5), there is little difference between the absorbed radiation in a sunspace for either published geometry with or without glazed end walls for those months typically requiring a heating load. Since the FCHART/SLR program already predicts the absorbed radiation for geometry (a) with insulated end walls, the same radiation values can be applied to geometry (a) with glazed end walls. Graph (C-1) shows this to be the case. The effects of using this radiation model on attached sunspaces of geometry (a) are shown in Figures 1 and 3 in the main body of the report.

Geometry (b) with insulated end walls requires the application of Equation (C-3) to the sunspace radiation processor in FCHART/SLR. Since only two transparent surfaces exist, Equation (C-2) can be modified to find the absorbed radiation in the sunspace as:

$$\bar{I} = \alpha \, [I_{tran_s}(1 - F_{s \rightarrow v}) + I_{tran_v}(1 - F_{v \rightarrow s})] \qquad \text{(C-6)}$$

where:

$\bar{I}$ is the monthly radiation absorbed in the sunspace,

α is the average sunspace absorptance for the nontransparent surfaces,

I_{tran_s} and I_{tran_v} are the monthly values of the radiation transmitted through the sloped and vertical south-facing surfaces respectively, and

$F_{s \rightarrow v}$ and $F_{v \rightarrow s}$ are the view factors from the sloped to vertical and vertical to sloped glazing surfaces, respectively.

This modification was incorporated into the radiation processor used by FCHART/SLR and run for the same 15 locations used previously. A typical run, again for Atlanta, Georgia, is shown in Figure C-4. As can be seen, the agreement between the radiation model using Equation (C-6) to determine the absorbed radiation and the LANL correlation is much improved. This adaptation to the radiation processor has been incorporated into FCHART/SLR and, as with the single south glazed surface, is used to estimate the absorbed radiation for geometries of type (b) for both glazed and insulated end walls.

The inclusion of Equation (C-6) into the FCHART/SLR program requires the determination of view factors $F_{s \rightarrow v}$ and $F_{v \rightarrow s}$. A subprogram VFF2 used by the PASOLE program to determine the view factor between adjacent surfaces has been added to the FCHART/SLR program. The inputs into this subprogram are the common length of the adjacent surfaces, the depth of each surface, and the angle between the two surfaces. Since one surface is assumed to be vertical and the slope of the other was already an input variable, the included angle is determinable. However, only the total areas of the two adjacent glazing surfaces are given in the passive input list of the FCHART/SLR program. Since the glazed end wall areas are no longer used in the radiation processor and a passive variable (Pn 27) can be used to include all end wall thermal losses, the end wall area variable (Pn 25) can then be changed to the sunspace width (i.e., the length of the adjacent glazed surfaces). This variable and the areas of the glazed surfaces determine the surface depth. The view factor from the sloped to vertical glazed surfaces is then found using the subprogram VFF2. This is then related to the view factor from the vertical to the sloped surface by:

$$F_{v \to s} = F_{s \to v} A_s/A_v \qquad \text{(C-7)}$$

where:

$F_{v \to s}$ and $F_{s \to v}$ are as previously defined, and

A_s and A_v are the sloped and vertical glazing areas respectively.

These view factors, passed as variables VFS (ISYS) and VFV (ISYS), are then incorporated into the radiation processor as indicated by Equation (C-6). Figure 3 in the main body of the report shows the type of agreement between the LANL sunspace FCHART/SLR program using this modification to the absorbed radiation.

Secondary differences between the two models include the effects of using monthly average values of glazing transmittances in the FCHART/SLR program rather than hourly values used by the PASOLE program. Also, the FCHART/SLR program neglects the effects of internally reflected radiation, implicitly assuming that all reflected radiation is lost. Both of these effects are small for the sunspace geometries published by LANL (reference 5). The monthly average transmittance used by the FCHART/SLR radiation processor accounts for a non-constant glazing transmittance (reference 6). The high opaque surface absorptance of those surfaces most likely to be directly illuminated by the sun, the floor and rear storage wall, also reduces any discrepancy induced by neglecting diffuse reflection between opaque surfaces in the sunspace. The net result of neglecting these two effects in the FCHART/SLR radiation model is to produce slightly conservative estimates of absorbed radiation.

The changes in the computer code for FCHART/SLR due to the inclusion of this modification are shown in Listing C-1.



Geometry (b) with Insulated End Walls



Geometry (a) with Insulated End Walls

FIGURE C-1 LANL Attached Sunspace Geometries



FIGURE C-2 Comparison of FCHART/SLR sunspace absorbed radiation results with published LANL correlations for sunspace geometry (a).



FIGURE C-3 Comparison of original FCHART/SLR sunspace absorbed radiation results with published LANL correlations for sunspace geometry (b).



FIGURE C-4 Comparison of FCHART/SLR sunspace absorbed radiation results after optical losses are included with the LANL correlations for geometry (b).

LISTING C-1 Modification of FCHART/SLR program to accept view factor requirements for multiple south-facing glazings on attached sunspaces. Changes made to subroutine PCALSO given in Appendix B.

Coding added to subroutine CALC

```
      FUA=0.0
      IF (XLIST(47) .EQ. 1.0) GOTO 220
C  CALCULATE VIEW FATORS FOR PASSIVE SYSTEMS  6-5-81 JMH
      DO 80 ISYS=1,NSYS
      VFV(ISYS)=0.0
      VFS(ISYS)=1.0
      IF(IFIX(PLIST(ISYS,1)+0.01).NE.1)GO TO 80
      IF(PLIST(ISYS,2).LT.3.5) GO TO 80
      IF(PLIST(ISYS,24).LT.1.0) GO TO 80
      C=PLIST(ISYS,25)
      A=PLIST(ISYS,23)/C
      B=PLIST(ISYS,24)/C
      TH = PLIST(ISYS,7)+90.
      VFSV=VFF2(A,B,C,TH)
      VFVS=VFSV*A/B
      VFS(ISYS)=1.0-VFSV
      VFV(ISYS)=1.0-VFVS
   80 CONTINUE
C  COMPUTE TOTAL PASSIVE COST, AREA, AND UA ACROSS ALL SYSTEMS
      DO 140 ISYS=1,NSYS
```

Changes in passive input list

Original

```
P1 23 AREA (SUNSPACE: OF TILTED SURFACE)...........    350.00 FT2
P1 24 SUNSPACE ONLY: VERTICAL AREA..................      0.00 FT2
P1 25 SUNSPACE ONLY: ENDWALL AREA (EACH)............     70.00 FT2
P1 26 SUNSPACE ONLY: U VAL OF SEPARATING WALL......       .10
P1 27 SUNSPACE ONLY: NON-GLAZING UA VAL.............     50.00 BTU/F-HOUR
P1 28 SUNSPACE ONLY: GLAZED ENDWALLS? 1=YES 2=NO...       2.00
```

Revised

```
P1 23 AREA (SUNSPACE: OF TILTED SURFACE)...........    350.00 FT2
P1 24 SUNSPACE ONLY: VERTICAL AREA..................      0.00 FT2
P1 25 SUNSPACE ONLY: SUNSPACE WIDTH ................     30.00 FT
P1 26 SUNSPACE ONLY: U VAL OF SEPARATING WALL......       .10
P1 27 SUNSPACE ONLY: NON-SOUTH GLAZING UA VAL......      50.00 BTU/F-HOUR
P1 28 SUNSPACE ONLY: GLAZED ENDWALLS? 1=YES 2=NO...       2.00
```

Note: Passive list 27 now includes all end-wall thermal losses. A U-value of 0.55 BTU/hr°F ft2 is assumed for the south-facing glazing

LISTING C-1 (continued)

Additional subfunction VFF2 from the PASOLE program

```
      FUNCTION VFF2 (A,B,C,TH)
C -- VIEW FACTOR FROM 1 TO 2. RECTANGULAR SURFACES WITH COMMON EDGE
C -- A,B ARE DIMENSIONS OF SURFACES 1 & 2 PERPENDICULAR TO
C -- COMMON EDGE, C IS DIMENSION ALONG COMMON EDGE (SAME FOR
C -- BOTH SURFACES), TH IS ANGLE BETWEEN SURFACES (DEG)
C
      DATA PI /3.141592654/
      T1=TH*PI/180.
      T2=2.*T1
      SINT=SIN(T1)
      COST=COS(T1)
      SINT2=SINT**2
      SIN2T=SIN(T2)
      COS2T=COS(T2)
C
      X=B/C
      Y=A/C
      X2=X**2
      Y2=Y**2
      Z=X2+Y2-2.*X*Y*COST
      Z1=SQRT(Z)
      Z2=SQRT(1.+X2*SINT2)
C
      VFF=.25*(2.-SINT2)*ALOG((1.+X2)*(1.+Y2)/(1.+Z))+.25*SINT2*(Y2*ALOG
     1 (Y2*(1.+Z)/(1.+Y2)/Z)+X2*ALOG(X2*(1.+X2)**COS2T/(Z*(1.+Z)**COS2T)
     2 ))+Y*ATAN(1./Y)+X*ATAN(1./X)-Z1*ATAN(1./Z1)
C
      IF (TH.EQ.90.) GO TO 1000
C
      VFF=VFF+SIN2T*.25*(2.*SINT*X*Z2*(ATAN(X*COST/Z2)+ATAN((Y-X*COST)
     1 /Z2))-X*Y*SINT-(PI/2.-T1)*(X2+Y2)-Y2*ATAN((X-Y*COST)/Y/SINT)-X2
     2 *ATAN((Y-X*COST)/X/SINT))
      N=50
      NP=N+1
      DG=Y/FLOAT(N)
      DFI=0.
C
      DO 200 I=1,NP
      FAC=2.
      IF (I.EQ.1.OR.I.EQ.NP) FAC=1.
      G=DG*(I-1)
      H=SQRT(1.+G**2*SINT2)
      DFI=DFI+FAC*H*(ATAN((X-G*COST)/H)+ATAN(G*COST/H))
  200 CONTINUE
C
      VFF=VFF+.5*DG*COST*DFI
C
 1000 CONTINUE
C
      VFF2=VFF/PI/Y
      RETURN
      END
```

APPENDIX D

Detected Coding Errors in FCHART/SLR

1. Subroutine SHADE

   Original

```
C..AUTOMATED SHADING ALGORITHM, SHAD=-1
   10 N=IDAYS(MONTH)
      DECL=23.45*SIN(D2R*360.0*(284.0+FLOAT(N))/365.0)
      ALTH=R2D*ACOS(SIN(D2R*DECL)*SIN(D2R*ALAT)+COS(D2R*DECL)*COS(D2R*AL
     1AT))
      ALTV=90.0-ALTH
      DECLJU=23.085911
      DECLDE=-23.332521955
      ALTHDE=R2D*ACOS(SIN(D2R*DECLDE)*SIN(D2R*ALAT)+COS(D2R*DECLDE)*COS(
     1D2R*ALAT))
      ALTVDE=90.0-ALTHDE
      ALTHJU=R2D*ACOS(SIN(D2R*DECLJU)*SIN(D2R*ALAT)+COS(D2R*DECLJU)*COS(
     1D2R*ALAT))
      ALTVJU=90.0-ALTHJU
C        FALTV=(ATAN(D2R*ALTV)-ATAN(D2R*ALTVDE))/(ATAN(D2R*ALTVJU)-
C        1ATAN(D2R*ALTVDE))
      FALTV=(ATAN(D2R*ALTV)-ATAN(D2R*ALTVDE))/(ATAN(D2R*ALTVJU)-ATAN(D2R
     1*ALTVDE))
```

   Revised

```
C..AUTOMATED SHADING ALGORITHM, SHAD=-1
   10 N=IDAYS(MONTH)
      DECL=23.45*SIN(D2R*360.0*(284.0+FLOAT(N))/365.0)
      ALW = 1.0/ATAN(D2R*(ALAT+23.33255))
      ALS = 1.0/ATAN(D2R*(ALAT-23.31441))
      AL = 1.0/ATAN(D2R*(ALAT-DECL))
      FALTV = (AL-ALW)/(ALS-ALW)
```

2. Subroutine PCALSF1

   Variables LCR and LCRs should be declared as Real.

   The key used to select the proper sunspace correlation should be as follows:

   KEY = 3 - 2 * (IEFLAG -1) + NIFLAG

   The BLC used to determine the variable XKAY is unit dependent and should be multiplied by RCONV, or:

   XKAY = 1.0 + UP(KEY) * PAREA / RCONV / BLC